
Abstract. This paper discusses several different ways to model the wellknown gossiping girls problem in promela. The highly symmetric nature
of the problem is exploited using plain promela, topspin (an extension
to Spin for symmetry reduction), and by connecting Spin to bliss (a tool
to compute canonical representations of graphs). The model checker Spin
is used to compare the consequences of the various modelling choices.
This – tutorial style – paper is meant as a road map of the various ways
of modelling symmetric systems that can be explored.

Introduction

In the early 1970s, the following puzzle (popularized by Paul Erdös) was circulated among mathematicians [9]:
There are n girls, each of whom knows a unique piece of initial information. They communicate by telephone calls, and whenever two speak
they share all the gossip that they know. The goal is to determine the
minimum number of calls necessary for all of the girls to learn all of the
initial information.
A number of researchers have independently proven that 2n−4 calls are necessary
and sufficient to achieve the goal. See [9] for an overview. A solution for 2n − 4
is straightforward. From [6]: “For n ≥ 4, 2n − 4 conversations suffice. For four
persons A, B, C and D, say, take conversations AB , and CD, followed by AC and
BD. For every additional person P , schedule one conversation AP , before A, B,
C and D interchange their knowledge, and another conversation AP afterwards.”
Given the optimal 2n − 4 solution, it is clear that we are not really interested
in using a model checker to compute this optimal solution. But given the highly
symmetric nature of this problem, it is interesting to see whether the model
checker Spin can cope with this. And if not, what options do we have to improve Spin’s performance? We also want to explore different ways to model this
problem. The first thing that comes to mind is to model each girl by a process
and a telephone call by sending and receiving messages. But, as we will see, there
are many ways to model the exchange of gossips between the girls.
Using the gossiping girls problem as a running example, we will show how
a symmetric model can be analyzed with Spin. Our goal is to inspire other
verification engineers – when facing modelling challenges – to explore alternative

modelling routes, especially in the case of a problem with apparent symmetries.
The paper tries to retain the tutorial style of presentation of [13,15] to make
the techniques easy to be adopted by intermediate to advanced Spin users. The
effectiveness of the techniques is illustrated by some experiments.
Related Work. In [17], Frits Vaandrager uses the gossiping girls problem to introduce the uppaal model checker. In [8], Joost-Pieter Katoen describes the
gossiping girls problem as an example of gossiping networks. The paper itself focuses on the performance evaluation of such networks. Curiously, Katoen writes
in a footnote: “The solution to instances of the gossiping girls example, e.g., can
easily be computed using off-the-shelf model checkers in a few seconds.” This
might be true for small instances of the problem, i.e., n ≤ 6, but does not scale
up for slightly larger n, as we will see in this paper.
In [11], Arend Rensink reports on experiments with the groove tool set on
various highly symmetrical problems, including the gossiping girls problem. We
will discuss groove in more detail in Sec. 6, where we compare our results with
groove.
Two extensions to Spin which exploit symmetry reductions for promela
models should be mentioned: symmspin [1], a tool which lets the user specify
symmetries through scalarsets and topspin [2,23], a tool which can automatically detect symmetries in promela models. In Sec. 4, we will use topspin on
a promela model of the gossiping girls.
Overview. Sec. 2 presents two straightforward promela specifications which
model the problem. We also explain how to obtain the optimal solution to the
problem with Spin. In Sec. 3 we report on our most optimal model in vanilla
promela, i.e., without using external tools and/or embedded C code. Sec. 4
discusses a promela model which we feed to topspin. In Sec. 5 we use Spin’s
embedded C extensions to connect our promela model to bliss, a tool which
computes canonical representations of graphs. Sec. 6 discusses the experiments
that we performed with Spin. Sec. 7 concludes the paper.
Source code. This paper presents several promela specifications which model
the gossiping girls problem. All promela models consider the problem for n =
4 girls. Of course, the models that we have used for our benchmark experiments have been parameterized in n, i.e., using the m4 macro processor [16].
All promela models and the verification results as discussed in this paper are
available from http://ruwise.nl/spin-2012/.

Initial Attempts

In this section we will discuss two straightforward approaches to model the
gossiping girls problem in plain promela. We will also explain how to find the
minimal solution of the problem with Spin.
Girl processes Fig. 1 shows the promela model girl-processes. The proctype Girl models the behavior of a single girl. The knowledge of a girl i is

byte knows[4] ;
chan phone = [0] of { byte } ;
byte calls ;

/* bit vector of gossips
*/
/* phone connections
*/
/* number of telephone calls */

proctype Girl(byte x) {
byte y;
do
:: atomic { phone ! x }
:: d_step { phone ? y ->
if
:: knows[x] != knows[y] ->
knows[x] = knows[x] | knows[y];
knows[y] = knows[x];
calls++;
:: else
fi;
y=0;
}
od
}
init {
atomic {
knows[0]=1<<0; knows[1]=1<<1;
knows[2]=1<<2; knows[3]=1<<3;
run Girl(0); run Girl(1); run Girl(2); run Girl(3);
}
}

Fig. 1. Promela model girl-processes.

stored in the global bitvector knows[i]. If the j-th position (from the right) in
knows[i] is set, it means that girl i knows the j-th gossip. Within init, the
initial gossips are set using the bitshift-operator <<.
Each girl uses the channel phone to call one of the other girls. When two girls
are connected, the mutual knowledge in knows[x] (the callee) and knows[y]
(the caller) is exchanged and the total number of calls is updated; but only
when new information has been exchanged. Exchanging the gossips themselves
is easy: a simple bitwise-or (|) of the bitvectors suffices.
Due to the interleaving semantics of Spin all possible sequences of calls between the Girl processes will be considered.
Single process A drawback of the girl-processes model is that, for each
proctype instance, Spin will allocate memory in the state vector. This is not
really necessary as we are only interested in all possible configurations of the
knows array.
In the model single-process as listed in Fig. 2 we resolve this drawback
by exploiting Spin’s non-deterministic choice to generate all possible scenarios.
There is only a single process allgirls which contains a do-loop. Each choice
in the do-loop represents a call between two girls that do not yet share the same
knowledge about the gossips. In this model the array of bitvectors is named k.

byte k[4];
byte calls;

/* bit vector of gossips
*/
/* number of telephone calls */

active proctype allgirls() {
k[0]=1<<0; k[1]=1<<1; k[2]=1<<2; k[3]=1<<3;
do
:: d_step { k[0] != k[1] -> k[0]=k[0] | k[1];
:: d_step { k[0] != k[2] -> k[0]=k[0] | k[2];
:: d_step { k[0] != k[3] -> k[0]=k[0] | k[3];
:: d_step { k[1] != k[2] -> k[1]=k[1] | k[2];
:: d_step { k[1] != k[3] -> k[1]=k[1] | k[3];
:: d_step { k[2] != k[3] -> k[2]=k[2] | k[3];
:: else -> break
od;
}

k[1]=k[0];
k[2]=k[0];
k[3]=k[0];
k[2]=k[1];
k[3]=k[1];
k[3]=k[2];

calls++;
calls++;
calls++;
calls++;
calls++;
calls++;

}
}
}
}
}
}

Fig. 2. Promela model single-process.

Another drawback of girl-processes is that we cannot easily express and
exploit the fact that an exchange between girls i and j is not really different
from an exchange between girls j and i. The single-process model easily allows
to express this by always choosing girls i and j with i < j. This cuts away half
of the outgoing transitions of a state, and consequently the number of states
matched.
Finding the optimal solution If we feed one of the promela models to Spin,
a safety run will generate all possible states of the models. And due to Spin’s
smart static analysis, the statement calls++; and the variable calls will even
be removed from the model, as the variable calls is never used in the model.1
We can also use Spin to find an optimal sequence of calls in the sense that
the number of calls is minimal [15]. From Sec. 1 we know that for this optimal
sequence the following holds: calls == 2*N-4. For the Promela models, we
have defined a boolean expression ALLKNOW which is true when all girls know all
gossips. For example, for the single-process model, this expression could be
defined as follows (remember that n = 4, and 15 is decimal for 1111 binary):
#define ALLKNOW (k[0] == 15 && k[1] == 15 && k[2] == 15 && k[3] == 15)

The logical formula ‘ALLKNOW ⇒ calls >= 2*N-4’ is invariant and holds for
all states. To get an error trail to a state where calls == 2*N-4 and ALLKNOW
both hold we could define the following promela process:
#define P (ALLKNOW && calls == 2*N-4)
active proctype monitor() { atomic { P -> assert(!P)} }

1

One should not replace the statement calls++; by the semantically equivalent
calls=calls+1; though. Spin regards this latter statement as a usage of the
variable calls and will thus retain the variable and statement in the system. With
a blow-up of the number of states as a result.

The property P corresponds with the goal state. The process monitor is blocked
on this condition P. At the moment P becomes true, the atomic sequence can
be taken and the assert of the negation of P will trigger an error. If we had
used assert(!P) instead of the atomic sequence, this would have doubled the
number of states. See [14] for details.
The safety check using the proctype with the assert will potentially visit all
states, as Spin has to check the property for all states. We can do better though.
Instead of a safety check, we can also use Spin’s liveness mode and check the
following ltl property:
(!ALLKNOW) U (calls > 2*N-4)

The property corresponds with paths where condition ALLKNOW remains false
until we reach a state where calls > 2*N-4. This property will be violated on
paths where we reach a state where ALLKNOW is true and calls > 2*N-4 is false,
i.e., our goal state. The reason why this liveness property should be preferred
over the safety check is the following: for each execution path for which the
until-formula holds, Spin will stop searching as soon as it hits a state for which
call > 2*N-4 holds.
Care should be taken when choosing the options for checking this liveness
property: the until-operator in Spin is the strong until-operator. This means
that the property p U q is violated on an infinite path where q never becomes
true. In our promela models there are infinite paths where the property call
> 2*N-4 never becomes true: paths containing ‘calls’ where no new information
is being exchanged. For this reason, we should check the above ltl formula with
‘acceptance cycle checking’ disabled. Due to the monotonic nature of the variable
calls, it is safe to do so: the state space has the shape of a tree, with self-loops
for the calls that do not exchange information.
Preliminary Results. Table 1 shows the verification results of all promela models of this paper and will be discussed in full in Sec. 6. For now, we are only
interested in the first two columns of Table 1 which show the results of verifying
the girl-processes and single-process model. It is interesting to see that
the number of states grows very fast as the number of girls increase. Spin still
manages to generate all states for n ≤ 6. For greater n, the problem is infeasible
for Spin (when limited to 2GB of memory). This is remarkable as the problem
at hand seems rather simple.
It should be clear that there is a lot of symmetry in these models. In fact,
if we have found a solution, any permutation on the set of girls gives rise to
new solution that is isomorphic to the original one. The tool to exploit this is
symmetry reduction, which we will further explore in Sec. 4. Before we do that,
we explore another line of reasoning to abstract from girl identity.

3

Channel of knowledge

In the initial phase of our study, we constrained ourselves to ‘pure’ promela
models and ‘vanilla’ Spin. When experimenting with Spin trying to solve the

problem more effectively, we noticed that there are many ways to incorporate
some notion of symmetry reduction into the model. In this section we discuss
our best approach – in vanilla promela – so far. The central observation for
this model is that if girls i and j have the same knowledge,
– a conversation between girl i and girl j is not going to add anything
– a conversation between girl i and any girl k with different knowledge is the
same as a conversation between girl j and girl k
So we might as well model equivalence classes of girls with the same knowledge. If we do so, we only have to record the number of girls in the equivalence
class. Denoting the size of an equivalence class with knowledge k by size(k),
an exchange between girls with knowledge ki and kj then results in size(ki )--,
size(kj )-- and size(ki |kj ) = size(ki |kj )+2. From an error trace with Spin, we
can easily produce a trace of girls within each equivalence class.
Fig. 3 lists the promela model chan-groups. Central to this solution is
the channel groups. The channel groups is a compressed representation of the
bitvector array knows of girl-processes: groups contains tuples (knowledge, size).
The knowledge-part is the bitvector representation of the gossips. The size-part
specifies how many girls have this knowledge. For example, the tuple (5,2)
means that two girls have the knowledge 101. The channel groups is used as a
sorted channel. In this way we get a canonical representation of the knowledge
information.
As with single-process, the do-loop is responsible for generating all possible conversations between the girls. In the two select-statements2 , the variables
i and j are non-deterministically set to indices of elements within the groups
channel.
The subsequent invocations of the getgroup macro remove these tuples from
the channel and set the information of the tuples to (ki,si) and (kj,sj),
respectively. After exchanging the gossips, the most elaborate part of the process
is storing the new knowledge back into groups. Note that due to fact that
groups is sorted, kc will always be different from ki. When both the knowledge
associated with i and j have changed, we use Spin’s random receive operator (??)
to inspect whether there is already a tuple in groups which has the knowledge
kc. If this is the case, two girls will be added to this tuple. Otherwise, a new
tuple (kc,2) is added to groups.
Fig. 4 lists the definitions of the inline macros getgroup and remove. The
macro getgroup is hairy as we need to get the i-th element out of the sorted
fifo-channel groups. The first 0 . . . (ix − 1) elements of groups are copied to
the end of groups using Spin’s regular send operation. Then, the ix-th element
is stored into (kx,sx). Finally, the remaining elements are copied to the end of
groups. In this way we keep the channel sorted without the need of an extra
2

The statement select is new in Spin version 6. The statement select(i:
n..m); sets i non-deterministically to a value in the range n . . . m.

#include "chan-inlines.h"
chan groups = [4] of {short,byte};
byte maxgroup = 3;
byte calls;

/* fields: (knowledge, size)
/* is equal to len(groups)-1
/* number of calls

*/
*/
*/

/* local, temporary variables of the process Girls: */
hidden byte i, j, ii, ki, kj, si, sj, kk, sk, kc, sc;
proctype Girls() {
do
:: if
:: len(groups) == 1 -> break;
:: else -> atomic {
select(i: 0..maxgroup-1);
select(j: i+1..maxgroup);
getgroup(i,
ki, si);
getgroup(j-1, kj, sj);
kc = ki | kj;
if
:: kj == kc ->
/* only i has changed */
remove(ki,si);
groups !! kj,sj+1
:: else ->
/* i and j have changed */
remove(ki,si);
remove(kj,sj);
if
:: groups ?? [eval(kc),sc] -> /* existing group */
groups ?? eval(kc),sc;
groups !! kc,sc+2;
:: else ->
/* new group
*/
groups !! kc,2;
maxgroup++
fi
fi;
calls++;
i=0; j=0; ii=0; ki=0; si=0; kj=0; sj=0;
kk=0; sk=0; kc=0; sc=0;
}
fi
od
}
init {
groups !! 1<<0,1; groups !! 1<<1,1; groups !! 1<<2,1; groups !! 1<<3,1;
run Girls();
}

Fig. 3. Promela model chan-groups.

channel.3 The macro remove removes one girl from the tuple (kx,sx). That is,
if there are still remaining girls with knowledge kx, the tuple will be put back
with sx-1. If sx==1, nothing will be put back into groups.
3

For our experiments we replaced the two calls to getgroup in chan-groups by a
call to a more efficient (but even more verbose) macro getgroups which retrieves
both tuples in a single walk over groups.

inline getgroup(ix, kx, sx)
{
ii=0;
do
:: ii < ix -> groups ? kk, sk;
groups ! kk, sk; /* put at the end */
ii++;
:: ii == ix -> groups ? kx, sx;
do /* move remaining elements to end */
:: ii < len(groups) -> groups ? kk, sk;
groups ! kk, sk;
ii++
:: else -> break
od;
break;
od;
ii=0; kk=0; sk=0;
}
inline remove(kx, sx)
{
if
:: sx > 1 -> groups !! kx,sx-1
:: else
-> maxgroup-fi
}

Fig. 4. Include file chan-inlines.h.

Semantically, the hidden variables i . . . sc are local variables of the proctype
Girls. They are only used within the atomic clause to compute the new groups.
By defining these variables as hidden variables, they will not be stored in the
state vector, and thus reducing each state by 12 bytes.
The model chan-groups is more effective than the previous attempts in
the sense that it minimizes the outgoing transitions of states. Given two tuples (k1 , s1 ) and (k2 , s2 ) in groups, only a single conversation between a girl
with knowledge k1 and a girl with knowledge k2 will be considered. Whereas in
the corresponding state of the previous models, s1 × s2 conversations would be
considered.
Although this model outperforms the models of the previous section, it still
does too much work. Even though we have abstracted from the identity of girls
in the exchange part, we still are left with a symmetry in the identity of the
gossips exchanged, which is encoded in the knowledge bitvectors.

4

TopSPIN

In this section we will report on our ventures with topspin to solve the gossiping
girls problem. topspin [2,23] is an automatic symmetry reduction tool for the
Spin model checker. topspin is applied to a promela model, and – provided
that the model adheres to some restrictions – uses computational group theory
to automatically determine a group of component symmetries associated with

chan
chan
chan
chan
chan
chan
byte

phone =
k1
=
k2
=
k3
=
k4
=
tmp
=
calls;

[0]
[4]
[4]
[4]
[4]
[4]

of
of
of
of
of
of

{
{
{
{
{
{

chan };
pid }; /* ki = knowledge of Girl with pid i
pid };
pid };
pid };
pid }; /* temporary channel
/* number of telephone calls

proctype Girl(chan know) {
pid p;
bool newinfo;
chan friend = know;

/* outside of d_step, friend == know

*/

*/
*/

*/

know ! _pid;
do
:: atomic { phone ! know; }
:: d_step {
phone ? friend;
newinfo=false;
do /* 1. move elements from friend to tmp */
:: empty(friend) -> break
:: nempty(friend) -> friend?p; tmp!p;
if
:: know ?? [eval(p)] -> skip
:: else -> newinfo=true
fi
od;
do /* 2. move elements from know to tmp */
:: empty(know)
-> break
:: nempty(know)
-> know?p;
if
:: tmp ?? [eval(p)] -> skip
:: else -> tmp!p; newinfo=true;
fi;
od;
do /* 3. move elements from tmp to friend and know */
:: empty(tmp)
-> break;
:: nempty(tmp)
-> tmp?p; friend!!p; know!!p
od;
if
:: newinfo -> calls++
:: else
fi;
p=0; friend=know; newinfo=false;
}
od
}
init { atomic { run Girl(k1); run Girl(k2); run Girl(k3); run Girl(k4); } }

Fig. 5. Promela model symm-girls.

the specification. The tool automatically modifies the model checking algorithm
employed by Spin to exploit these symmetries during verification. We also considered the older symmspin package [1] but preferred topspin because the tool
is fully automatic and gives helpful explanations in cases when the tool cannot
find symmetries in the model.
topspin uses the gap [19] computational algebra system to effectively detect state space symmetry from the associated promela model. Furthermore,

topspin uses a prototype extension of the saucy program [21], which is used
to compute symmetries of directed graphs. As mentioned, topspin places some
restrictions on the promela model, including:
All processes should be instantiated using run statements within init.
The current version of topspin does not support arrays of channels.
topspin is not yet compatible with the verification of liveness properties.
For symmetry to be detected, it is important for proctypes to use their
built-in pid variable rather than a user-defined process identifier.
– topspin does not support bitvectors in the sense that it recognizes a byte
being used as an array from pid to bool.
–
–
–
–

Especially the last two restrictions forced us to rethink the representation of
knowledge within our model. Fig. 5 lists the promela model symm-girls. This
model is a variation of the original girl-processes model. The behavior of
each girl is represented by a Girl process. However, instead of using an arbitrary
number to identify the girls, we now use the process identifier (pid) of a Girl
process to identify the gossip that she originally knew.
Instead of using bitvectors to store the knowledge of the girls, we now encode
the knowledge in sorted channels of pid’s. For each Girl with pid i, a channel
ki is defined, which holds the pids of all gossips that this girl knows. When the
processes are instantiated in the init process, the channel ki is passed to the
Girl-process which will get pid i. This channel is stored in the Girl’s know
variable. Now, in combination with the symmetric version of the process Girl,
topspin will recognize that these channels can safely be permuted. As with
the chan-groups model, we keep these channels sorted to obtain a canonical
representation of the knowledge.
As before, the girls exchange information through the rendez-vous channel
phone. The girls offer their own knowledge channel know on this channel phone.
The receiving girl makes sure that knowledge is exchanged and that both her
own knowledge channel (know) and the knowledge channel of the other girl
(friend) get updated. Updating know and friend requires three steps. Firstly,
all elements of friend are copied to the temporary channel tmp. Secondly, all
elements of know which are not yet in tmp are copied to tmp. Now tmp contains
the combined knowledge of the original friend and know channels. Thirdly,
all elements of tmp are copied to friend and know using Spin’s sorted send
operation. Note that we only update the variable calls if the conversation
between the two girls revealed new information. At the end of the d step we
again reset all local variables.
The symmetric model for topspin comes with a price, though. With respect
to the information in the state vector, the symm-girls model is quite expensive,
compared to the previous models. Instead of using an efficient array of bitvectors,
n+1 channels of length of n are being used. Furthermore, exchanging information
has become an expensive operation. Instead of using a single bitwise operator,
a single conversation requires the copying of three of those channels.

c_decl { \#include "bitmatrix.h" }
c_code {
unsigned int automorph_partition[4];
BitMatrix stack_adjmat;
BitMatrix canonical_adjmat;
}
c_track "&automorph_partition" "sizeof(unsigned int)*4"
c_track "stack_adjmat.pbv"
"((4*4)/8+((4*4)%8>0))"
c_track "canonical_adjmat.pbv" "((4*4)/8+((4*4)%8>0))"

"StackOnly"
"StackOnly"

#define SINGLE_CALL(x,y)
c_expr { designated_partners(&stack_adjmat,
automorph_partition, x, y) } ->
calls++;
c_code {
merge_rows(&stack_adjmat, x, y);
comp_canonical(&stack_adjmat, &canonical_adjmat);
};

\
\
\
\
\
\
\

byte calls=0;
active proctype allgirls() {
c_code {
init_bit_matrix(&stack_adjmat, 4);
init_bit_matrix(&canonical_adjmat, 4);
};
do
::

c_code { comp_automorph_partition(&stack_adjmat,
automorph_partition); };
if
:: d_step { SINGLE_CALL(0,1) }
:: d_step { SINGLE_CALL(0,2) }
:: d_step { SINGLE_CALL(0,3) }
:: d_step { SINGLE_CALL(1,2) }
:: d_step { SINGLE_CALL(1,3) }
:: d_step { SINGLE_CALL(2,3) }
:: else -> break
fi;

od;
}

Fig. 6. Promela model bliss.

As mentioned above, the current version of topspin cannot yet deal with liveness properties. Therefore, for the verification runs to find the optimal solution
to the problem, we had to resort to the less effective assert statement.

5

Bliss

In this section we exploit Spin’s embedded C code extensions to encode the
symmetries ourselves. This section assumes a working knowledge of Spin’s C
code extensions, i.e., chapter of 17 of [3]. Basic knowledge of graph isomorphism
theory is also assumed. The approach taken for this model is based on the
abstraction techniques of [4].
The original promela models of Sec. 2 use an array knows of bitvectors
to store the knowledge within the system. This array knows is essentially the

adjacency matrix of a graph representation of the knowledge: if a girl i knows
the gossip of girl j, there is a directed edge from i to j. For this final promela
model, we exploit this graph representation. We will use the bliss tool to find
the canonical representatives for these graphs.
bliss [7,18] is a tool for computing automorphism groups and canonical forms
of graphs. It has both a command line user interface as well as C/C++ programming language apis. Compared to other well-known tools for checking graph
isomorphisms (e.g., nauty and saucy), bliss seems to perform quite well [7].
Two features of bliss which make the tool ideal for our purposes are the
following: (i) bliss can work with directed graphs, and (ii) bliss can compute a
canonical representation of the automorphism group of a graph. Given a graph
G and its canonical representative graph ρ(G) the following holds: for all graphs
G′ that are isomorphic with G, bliss will return the same representative graph,
i.e., ρ(G) = ρ(G′ ).
Fig. 6 shows the promela model bliss. The model is a variation of the
single-process model: there is a single process allgirls, which enumerates
all possible calls between the girls in a do-loop. We use an efficient bitmatrix to
store the knowledge of the girls. In fact, we use two bit matrices: stack adjmat
and canonical adjmat. The stack adjmat is the adjacency matrix of the directed graph that we work with in the body of the proctype allgirls. When a
conversation takes place between two girls, this adjacency matrix gets updated.
However, this adjacency matrix is never saved in the state space: it is only used
in the current state. Hence the decoration "StackOnly" next to the c track
declaration of stack adjmat. Instead of stack adjmat, we store the adjacency
matrix of its canonical graph: canonical adjmat. Note that canonical adjmat
does not have the "StackOnly" decoration within its declaration. In terms of
[4], canonical adjmat is the abstract graph.
Let us look at the macro SINGLE CALL. The knowledge of the two girls x and
y is merged using merge rows, which performs the familiar bitwise-or operation
on the rows of x and y. The function comp canonical computes the canonical
representation of stack adjmat. First, in this function, the adjacency matrix
of stack adjmat is converted to bliss’s internal graph representation. Then,
bliss’s function to compute the canonical representation of a graph is called.
Finally, the internal graph representation is converted back to our BitMatrix
representation.
An important part of SINGLE CALL is the call to designated partners.
Converting the adjacency matrix to bliss’ graph representation (and vice-versa),
and especially the computation of the canonical representation of a graph are
time-consuming procedures. To improve the running time of the verification
with Spin, it is important to eliminate, when possible, calls between girls that
are guaranteed to lead to a canonical graph that has or will be covered by
connecting two other girls. Given two potential partners x and y, the function
designated partners returns true, when:
– the knowledge of x and y is different
(i.e., the rows of the bitvectors x and y differ), and

– if x and y belong to the same cell of the partition of the automorphism
group, then x and y should both have the lowest indices of this cell, or
– if x and y belong to different cells of the partition of the automorphism
group, they both should have the lowest indices within their respective cells.
A partition of a set V is a set of disjoint non-empty subsets of V whose union is V .
The elements of a partition are called cells. Given a graph G, the automorphism
group Aut(G) of G can be represented by a partition P (G) of the nodes V of
G: the nodes in each cell of P (G) can be permuted. For example, in the initial
state of the model, when all girls only know their own gossip, the partition of the
automorphism group consists of a single cell containing all nodes. It is enough to
only consider a single conversation between two nodes of this cell. After the first
transition, the partition consist of two cells: cell c1 contains the two girls which
just exchanged their gossips and cell c2 contains the other girls. It is enough to
only consider a conversation between two nodes of c2 and a conversation between
a node of c1 and a node of c2 .
The partition is computed in comp automorph partition. Within this function, bliss’ find automorphisms function is called, which returns a set of generators for the automorphism group of the graph. Given this set of generators
the cells can be computed. In Fig. 6 these cells are stored in the "StackOnly"
array automorph partition. If two nodes n1 and n2 belong to the same cell,
automorph partition[n1 ] == automorph partition[n2 ].
Implementation. Although bliss has a well documented api, connecting the tool
to Spin took some considerable effort. Roughly 300 lines of C code were needed
for this interface.

6

Experiments

This section discusses the experiments that we conducted with Spin on the various promela models. During our study, we carried out two types of experiments
with Spin: ‘exhaustive’ experiments where we used Spin’s standard safety mode
to generate all reachable states for the given promela model, and ‘find the optimal sequence’ experiments where we used Spin’s liveness mode to find a 2n − 4
solution to the gossiping girls problem.
We only report on the ‘exhaustive’ experiments. The ‘find the optimal sequence’ experiments show similar results though; except for the fact that in
some cases Spin could still find the optimal sequence where the corresponding
‘exhaustive’ experiment would run out of memory. However, these results depend
on Spin’s default dfs exploration order. Since version 5.1.x, Spin supports several C compile time options (e.g., -DREVERSE, -DRANDOMIZE) which change the
order of selecting outgoing transitions. Running the ‘find the optimal sequence’
experiments with any of these options will thus show different results. Therefore we decided that including the results of the second type of experiments was
not very meaningful. Instead, we report on our experiments with the swarm
tool [5,22] to ‘find the optimal sequence’.

n=4

sv (bytes)
states, stored
states, matched
time (sec)
memory (MB)

64
190
2,080
0
131

20
192
670
0
131

44
140
83
0
131

104
251
2,391
0
131

104
42
394
0
131

20
45
33
0
131

n=5

sv (bytes)
states, stored
states, matched
time (sec)
rate (s/sec)
memory (MB)

72
9,153
173,889
0.08
114k
131

20
9,155
67,139
0.01
no data
131

44
2,468
3,659
0.01
no data
131

120
10,234
186,098
0.36
28k
132

120
540
9,897
0.06
9.0k
131

20
321
753
0.03
11k
131

n=6

sv (bytes)
states, stored
states, matched
time (sec)
rate (s/sec)
memory (MB)

52
136
91,348 1,150,478
252,039 32,818,752
0.8
70.5
114k
16k
137
272

136
18,316
525,940
4.8
3.8k
133

20
6,505
31,400
1.2
5.2k
131

n=7

sv (bytes)
states, stored
states, matched
time (sec)
rate (s/sec)
memory (MB)

216
216
> 13 e6 1,334,922
> 5.3 e8 54,525,101
1460
815
9.0k
1.6k
o.o.m.
325

20
279,985
2,201,294
87
3.2k
142

n=8

sv (bytes)
states, stored
states, matched
time (sec)
rate (s/sec)
memory (MB)

exhaustive, solution found
swarm, solution found
modelling time

80
20
1,092,474 1,092,476
31,681,718 13,311,818
13.4
3.0
81k
369k
197
172
88
> 28 e6
> 11.4 e8
541
52k
o.o.m.

20
52
> 37 e6 6,588,212
> 6.6 e8 29,214,527
195
96
191k
69k
o.o.m.
533
60
> 27 e6
> 17 e7
595
47k
o.o.m.

n≤7
n≤8
< 1 hour

n≤7
n≤8
< 1 hour

n≤7
n ≤ 10
≈ 4 days

240
20
> 11 e6 23,461,597
> 62 e7 267,185,640
14,300
11,100
0.8k
2.1k
o.o.m.
1026
n≤7
n≤7
n.a.
n≤6
≈ 2 days

n≤8
n ≤ 10
≈ 8 days

Table 1. Summary of exhaustive benchmark experiments.

Settings. All experiments were run on an Apple MacBook 2.66GHz Intel Core
2 Duo, with 8GB of RAM, running Mac OS X 10.6.8 (in single-user, console
mode) and gcc 4.2.1. We used Spin version 6.1.0 for all experiments, except for
the topspin experiments for which we used Spin version 5.2.5. We used topspin
version 2.2.5, bliss version 0.50 and groove version 4.4.5. For the topspin
experiments we used topspin’s default fast symmetry reduction strategy (which
is indeed fast, but does not yield canonical representations). For each verification
run we reserved 2GB of memory. To compile Spin’s pan verifiers for safety runs,
we used the following options for gcc:
gcc -O2 -o pan -DSAFETY -DNOCLAIM -DMEMLIM=2048 pan.c

Furthermore, for topspin and bliss, we linked pan to the respective libraries, of
course. We executed all pan verifiers using the following command-line:

./pan -m50000 -c1 -w24

Due to -w24, pan will always reserve 128MB of memory for the hash table.
Effort. The last row of Table 1 shows an estimate of the modelling time for the
various promela models. Parameterizing the models in n (using the m4 macro
processor) and setting up all experiments took us more than a week. Finally,
all experiments can be executed in roughly 28 hours. All promela models and
verification results are available from http://ruwise.nl/spin-2012/.
Exhaustive experiments Table 1 lists the results of letting Spin compute
all reachable states for the various models. The columns correspond with the
promela models that have been discussed in this paper. The symm-girls model
has been verified with both vanilla Spin and with topspin.
The rows ‘time (sec)’ list the elapsed time as reported by Spin. The rows
‘rate (s/sec)’ list the number of states per second, again as reported by Spin.
When Spin does not have enough memory to finish the verification, this is indicated with ‘o.o.m.’ (out-of-memory). We still report the output by Spin though.
For topspin and bliss the ‘rate (s/sec)’ column needs clarification. Symmetry
reduction involves computing a canonical representative for every state that is
encountered. Every time topspin or bliss processes a state, the tool is really
processing a whole equivalence class of states.
From Table 1 we learn that our two initial promela models are fast. Especially, the rate of the single-process model is high, compared to the rates of
the other models. This is obvious as a transition in single-process only consists of a bitwise-or operation and two assignments. Moreover, the state vector
of single-process consists of only 20 bytes.
Our best vanilla promela model chan-groups behaves quite well. Compared to the two straightforward models, it succeeds in lowering the number of
states without becoming too slow. It succeeds for n = 7.
Due to n processes and the n + 1 channels of length n, the state vector of
the symm-girls model is large. We see that symm-girls exhibits poor results
when veryfying the model with vanilla Spin. Due to the expensive computation
involving the knowledge channels, the rate is also quite low. On the other hand,
in combination with topspin, the symm-girls model fares quite well for n ≤ 6.
The rate is even lower, but this is no surprise due to the computation of canonical
representatives. For n ≥ 7 this computational overhead becomes significant.4
The ‘winner’ of our experiments is the bliss model. It is the only model
which succeeds in generating all states for n = 8, albeit slowly. Spin needs
only slightly more than 1GB of memory to verify this model for n = 8. As
with the topspin model, we see that with increasing n, the verification rate for
this model deteriorates. This is not surprising given all the operations that are
executed within a single transition. Of these operations, with increasing n, the
automorphism detection becomes the most dominant factor.
4

We have discussed these verification results with Alastair Donaldson, the developer
of topspin. He suspects that the current version of topspin does not handle sorted
send channels in the most efficient way. A future version of topspin should fix this.

#girls
n=4
n=5
n=6
n=7
n=8

gossip-priorities
states
trans. time memory
55
253
0.3s
6kB
382
1,852
1.1s
0.5MB
4,458
22,492
4.0s
1.5MB
80,459
409,372 44.7s 17.1MB
4,157,679 22,684,164 3099s 899MB

gossip-nested
states
trans. time memory
13
66 0.2s
0.2MB
52
418 1.0s
0.3MB
361
4,749 3.2s
0.5MB
3,703
71,804 18.9s
2.6MB
62,982 1,790,230 519s 54.9MB

Table 2. Summary of experiments with groove.

Swarm swarm [5,22] is a tool, which – given a promela model and some verification settings – generates a script that performs many small verification jobs
with Spin in parallel, that can increase the problem coverage for very large verification problems. swarm’s verification jobs use Spin’s bitstate hashing mode.
We have used swarm on the promela models to ‘find the optimal sequence’
using the ltl property as discussed in Sec. 2. For each promela model (and for
each n ∈ {4, . . . , 10}) we first performed a test run with Spin to get an estimate
for the speed and the hash factor. These settings were then used to let swarm
run 60 minutes on each n-th version of the promela models. We let swarm
terminate as soon as one of the parallel verifications would find an error trail
(i.e., option -e). For n ≤ 6 swarm would come up with an error trail within a
few seconds. For n ≥ 7 it took longer.
The one-but-last row of Table 1 summarizes our experiments with swarm.
For both chan-groups and bliss, swarm succeeds in finding an optimal sequence for n = 10. Curiously, on symm-girls, the combination swarm and
topspin only finds an error for n ≤ 6. Perhaps the large state vector in combination with the slow speed are responsible for this. Obviously, if swarm would
have been given more time, it would probably have found more errors trails.
This experiment just illustrates the power of swarm in dealing with promela
models where exhaustive verification is not an option.
Groove As mentioned in Sec. 1, the groove tool set [10,20] has also been used
for experiments with the gossiping girls problem [11]. groove [20] is a project
centered around the use of simple graphs for modelling several structures objectoriented systems, and graph transformations as a basis for model transformation
and operational semantics. The groove tool set includes an editor, a simulator,
a generator for automatically exploring state spaces and a model checker.
One of the key features of the groove tool set is its underlying, very efficient
algorithm for isomorphy checking of graphs. With a symmetric model, such as
the one for the gossiping girls, you get your symmetry reductions for free.
On the same machine that we ran our Spin experiments, we have also ran the
groove generator to compute the state space. We allocated 2GB of memory for
the jvm and used groove’s dfs mode. Table 2 shows the results of the groove
experiments. The results for gossip-priorities correspond with groove models
which use rule priorities [11], whereas the gossip-nested results correspond with
improved groove models which use a rule system with nested rules [12]. This

rule system with nested rules has recently been added to groove. We see that
groove does not have any problems with n = 7 and clearly outperforms our
promela models for the exhaustive experiments.

7

Conclusions

We have explored several ways to model the gossiping girls problem along different points of view, trying to exploit the inherit symmetries of the problem.
The main point of the paper is that alternative modelling routes with Spin can
lead to substantial better verification results.
We have seen that straightforward promela models are easy to construct
and show reasonable results, but are not very efficient in the terms of the number
of states. However, we have also seen that even in vanilla promela we can implement symmetry reductions to lower the number of states. Remember though
that it took several modelling efforts to come up with the final chan-groups
model that we discuss in this paper.
topspin is a fully automatic, user-friendly tool, but it requires some basic
knowledge of group theory. And although topspin outperforms vanilla Spin
on the gossiping girls problem, for larger n one has to have patience. topspin
requires a truly symmetric model. This might mean that you either have to
rework your model or that you have to identify the symmetries yourself.
Although our promela model which uses bliss shows the best results, it took
quite some programming effort to get there. For the occassional symmetric model
this route is therefore not recommended. However, whenever a very memory
efficient model is required, this approach could be an option.
Due to its efficient algorithm for checking isomorphy, groove performs best
on the gossiping girls problem. The groove tool set also includes ctl and ltl
model checkers. So, if you have to verify a highly symmetric model – and you
are not afraid to learn a new modelling formalism – groove should be high on
your list of potential candidates.
Finally, we witnessed the power of swarm. The tool succeeds in finding the
optimal sequence in promela models where exhaustive verification surely would
have failed. The results are not uniform for all promela models though. More
research with swarm is needed.
We also intend to apply our modelling approach to other, larger symmetric
problems to see in howfar our results can be generalised. Furthermore, we intend
to extend our comparison with other model checkers.
Acknowledgements We would like to thank Arend Rensink for providing us
with the groove models of the gossiping girls problem and with his assistance
in running groove. Gerard Holzmann is thanked for his help with swarm. We
would like to thank Joost-Pieter Katoen and especially Alastair F. Donaldson
for their valuable comments and suggestions to improve the paper.

18

Theo C. Ruys and Pim Kars

References
1. D. Bosnacki, D. Dams, and L. Holenderski. Symmetric Spin. STTT, 4(1):92–106,
2002.
2. A. F. Donaldson and A. Miller. A Computational Group Theoretic Symmetry
Reduction Package for the Spin Model Checker. In M. Johnson and V. Vene,
editors, Proc. of AMAST 2006, LNCS 4019, pages 374–380. Springer, 2006.
3. G. J. Holzman. The SPIN Model Checker – Primer and Reference Manual.
Addison-Wesley, Boston, USA, 2003.
4. G. J. Holzmann and R. Joshi. Model-Driven Software Verification. In S. Graf and
L. Mounier, editors, Proc. of SPIN 2004, LNCS 2989, pages 76–91. Springer, 2004.
5. G. J. Holzmann, R. Joshi, and A. Groce. Tackling Large Verification Problems
with the Swarm Tool. In K. Havelund, R. Majumdar, and J. Palsberg, editors,
Proc. of SPIN 2008, LNCS 5156, pages 134–143. Springer, 2008.
6. C. Hurkens. Spreading gossip efficiently. Nieuw Archief voor Wiskunde, 5/1(2):208–
210, June 2000.
7. T. Junttila and P. Kaski. Engineering an Efficient Canonical Labeling Tool for
Large and Sparse Graphs. In D. Applegate, G. S. Brodal, D. Panario, and
R. Sedgewick, editors, Proc. of ALENEX 2007, pages 135–149. SIAM, 2007.
8. J.-P. Katoen. How to Model and Analyze Gossiping Protocols? acm sigmetrics’
Performance Evaluation Review, 36(3):3–6, Dec 2008.
9. D. Liben-Nowell. Gossip is Synteny: Incomplete Gossip and the Syntenic Distance
between Genomes. Journal of Algorithms, 43(2):264–283, 2002.
10. A. Rensink. The groove Simulator: A Tool for State Space Generation. In J. L.

